

ファイル・フォーマット変換プログラム “CONVRT. N88”
および γ 線スペクトル表示プログラム
“RIDISPLAY”

若林 文高

国立科学博物館 理化学研究部

A File Converting Program “CONVRT. N88” and a γ -ray
Spectrum Display Program “RIDISPLAY”

By

Fumitaka WAKABAYASHI

Department of Physical Sciences, National Science Museum, Tokyo

Abstract

A file converting program and a γ -ray spectrum display program are presented. These are the N₈₈-DISK BASIC programs for the researchers who use the facilities of Tokai Branch, Research Center for Nuclear Science and Technology, the University of Tokyo. At the Tokai Branch, they use the γ -ray spectrometer system consisting of a γ -ray detector, a preamplifier, a multichannel analyzer (TN-1750 or TN-7200, Tracor Northern), and a 16 bits personal computer MULTI-16 (Mitsubishi Electric Co. Ltd.). The data counted by the multichannel analyzer are transferred to the MULTI-16 through an RS-232C serial ASCII I/O port, and stored in an 8 inch floppy disk. All these steps are controlled by the operating system CP/M-86 used in MULTI-16.

The file converting program “CONVRT. N88” converts the CP/M-86 format file to the N₈₈-DISK BASIC format file. Once a disk is converted, it is possible to analyze the data, by PC-8801, PC-9801 (NEC Co. Ltd.) or other personal computers.

The γ -ray spectrum display program “RIDISPLAY” is one of the data processing programs for converted disks. It has functions for the spectrum display on CRT, data smoothing, and print-out of the numerical data. Some spectra processed by “RIDISPLAY” are shown in this report.

1. はじめに

パーソナル・コンピューター (パソコン) の発達・普及は、パソコンを用いた計測システムやデータの記憶・処理システムの発達をうながした。日本原子力研究所東海研究所内にある東京大学原子力研究総合センター原研施設共同利用部門東海分室では、 γ 線測定に次のようなシステムを採用している。すなわち、Ge (Li) γ 線検出器からの信号を増幅し、マルチチャンネルアナライザーで積算する。積算

終了後、測定データをデータバス RS-232C を通じて 16 ビット・パーソナルコンピュータに伝送し、8 インチフロッピーディスクに記憶させる。¹⁾このようにして得られたディスクを利用して各種データ処理が可能である。こうしたデータ伝送からデータ処理までの一連の操作は、オペレーティングシステム CP/M-86 の上で走るプログラムによって行われる。

得られたデータディスクを各研究室所有のパソコンでデータ処理できれば、測定の効率化、データ処理の迅速化を図ることができる。しかし、フロッピーディスクへのデータ記録のフォーマットは使用機種・使用言語によって異なるため、異機種・異言語のパソコンを用いる場合は、データの読み出しに何らかの工夫を必要とする。

そこで他機種パソコンを用いてデータ処理が可能なプログラムを作成したので報告する。プログラムは、フォーマット変換プログラム (CONVRT. N88)、スペクトル表示プログラム (RIDISPLAY)、データ読み取りプログラム (READPRGM) の 3 種から成る。ここでは NEC PC-8801 (又は PC-9801) を用いたが、一部を修正することにより、他機種にも適用できる。

2. 適用機種等

適用機種 NEC PC-8801 又は PC-9801

周辺装置 8 インチフロッピーディスクドライブ (2 ドライブ)、プリンター、CRT

使用言語 N₈₈-DISK BASIC

データ解析に用いるオリジナルディスクは、東京大学原子力研究総合センター原研施設共同利用部門東海分室の三菱マルチ 16 によってデータが書き込まれた 8 インチフロッピーディスクである。マルチチャンネルアナライザーとして Tracor Northern 社製 TN-1750 又は TN-7200 を使用した場合に適用できる。

3. プログラムの概要

フォーマットの異なるディスクをデータ処理する場合、次の 2 つの方法が考えられる。

1. オリジナルディスクから直接データを読み取る。
2. フォーマットを変換して新しいディスク (ここでは「変換ディスク」と称す) を作成し、データ処理にはこの変換ディスクを用いる。²⁾

前者は、フォーマット変換の手間が省け、オリジナルデータをそのまま活用できるという利点があるが、データ読み取りプログラムが複雑となる。後者は、新しいディスクを用意してフォーマット変換をする煩わしさがあるが、一度変換ディスクを作成したのちは、読み取りが容易となる。ここでは後者の方法を採用することとした。

一連の作業を図示すると Fig. 1 のようになる。

まず、変換プログラム “CONVRT. N88” によってオリジナルディスクから PC-8801 (又は PC-9801) 用の変換ディスクを作成する。次にスペクトル表示プログラム “RIDISPLAY” をランさせると、自動的に入力プログラム “READPRGM” がロード・ランされ、変換ディスクよりデータを CPU 内のメモリーに読み込む。読み込みが完了すると自動的に “RIDISPLAY” に戻り、このプログラムでは、スペクトルの CRT・プリンター出力、数値データの CRT・プリンター出力、スムージングなどのデータ処理が行える。

1 ファイルのチャンネル数は、利用頻度の高い 2048 ch に固定しており、チャンネル数を変える場合は、“CONVRT. N88”、“READPRGM” の一部を変換する必要がある。




Fig. 1. The flow chart for file converting, CRT display and other data processing used in this work.

各プログラムの概要を以下に記す。使用法およびプログラムリストは付録に記す。

3-1. フォーマット変換プログラム“CONVRT.N88”の概要

マルチ16 CP/M-86 フォーマットのファイルを N₈₈-DISK BASIC フォーマットのファイルに変換して新しいディスクを作成するためのプログラムである。次の 3 機能をもつ。

(1) ディスクアロケーションマップの読み取り³⁾

ファイルがディスク上のどの位置に記録されているかをブロック番号で示す。データ読み取り・データ転送では、このブロック番号で目的のファイルを指示する。

(2) データ読み取り

オリジナルディスク (マルチ16 CP/M-86 フォーマット) から直接データを読み取り、CRT 又はプリンターに出力する。読み取るブロック数を指示することにより、1 ファイルが 2048 ch 以外の場合も読み取りが可能となっている。

(3) データ転送

オリジナルディスクのファイルを N₈₈-DISK BASIC フォーマットに変換して別のディスクに転送する。1 ファイルを転送する時間は、PC-8801 でおおよそ 60 秒である。転送先のディスク (変換ディスク) は、データ専用としておく。

3-2. スペクトル表示プログラム“RIDISPLAY”の概要

基本的にはスペクトルを CRT 上に表示し、プリンターに出力するプログラムであるが、次のような機能をもつ。

(1) スペクトルの CRT 表示、プリンター出力

X 軸・Y 軸方向の拡大・縮小および平行移動が可能である。画面上では Fig. 2 のように表示される。画面右上方に各ファンクションキーの機能、左上方にデータファイル名、下方に X 軸・Y 軸の拡大率、ライブタイム、デッドタイムが表示される。プリンターでハードコピーをとると Fig. 3, 4 のようにファンクションキーの機能説明を除いて出力される。

スペクトルの表示方法は、“セン”、“テン”の 2 種が選択でき、“セン”では $XE < 2$ の場合、Fig 2 のように各点が直線で結ばれる。 $XE \geq 2$ では Fig. 3 のような表示となり、各縦線は 1 ch に対応し、ピーク幅を調べるのに有利である。“テン”では、Fig. 4 のように各 ch ごとに点でプロットされる。

(2) スムージング

3 点の平均をとり、スムージングを行う。PC-8801 でおおよそ 90 秒間を要す。スムージングを 1 回行った例を Fig. 5-(b) に示す。Fig. 5-(a) がスムージングを行う前である。

(3) 数値データの CRT 表示、プリンター出力




Fig. 2. An example of CRT display of γ -ray spectrum. The sample is Qingzhen enstatite chondrite, which was irradiated by thermal neutron after radiochemical separation.⁴⁾ And this Fig. is an example of line plot (f6) in the case of $XE < 2$.




Fig. 3. An example of line plot in the case of $XE \geq 2$. Other data are as those of Fig. 2.




Fig. 4. An example of point plot (f7). The data of this Fig. are the same as those of Fig. 2.

以上のように“RIDISPLAY”ではスペクトル表示のための基本的な機能が備えられたが、今後はデータ処理プログラムの骨格として、さらに拡充させていく予定である。

3-3. データ読み取りプログラム“READPRGM”

“RIDISPLAY”用の入力プログラムである。“CONVRT. N88”によって作成された変換ディスクからデータを読み取り，“RIDISPLAY”へと引き継ぐ。

4. 展 望

今回作成された“CONVRT. N88”、“READPRGM”により、変換ディスクを作成し、その変換ディスクからデータを読み取るまでの過程は確立した。したがって、今後はデータをいかに活用していくかがプログラム開発の要点となる。“RIDISPLAY”には第一段階として、スペクトル表示、スムージングや数値データ出力などの機能をもたせたが、今後はこれを骨格として、ピーク面積の計算(統計的な処理を含めて)、半減期補正、ピーク同定などの機能を付加して拡充してゆく必要がある。これらの機能は、いん石などの貴重な試料から多様な元素を、特にごく微量にしか含まれない元素を分析する際に有効な機能となってくる。

5. 謝 辞

本研究の一部は、昭和58年度および59年度文部省「日本原子力研究所原子炉等共同利用一般研究」の費用による。




Fig. 5. An example of smoothing (f_5) effects on γ -ray spectrum. The sample is Qingzhen enstatite chondrite after radiochemical separations.⁵⁾

(a) Before smoothing.

(b) After smoothing.

本プログラムを作成するにあたり有益な助言と示唆を与えてくださった理化学研究所海洋計測研究室杉原滋彦博士，当館理化学研究部島正子主任研究官に感謝いたします。また，東京大学原子力研究総合センター原研施設共同利用部門東海分室で共同利用の手続き，測定などの面でお世話していただいている伊藤泰男助教授，高野武美技官をはじめとするスタッフの方々に感謝いたします。

注および文献

- 1) a) Laboratory Equipment Corporation “TN-1750 INTERFACE UTILITY Version 1.02”
b) Laboratory Equipment Corporation “TN-7200 INTERFACE UTILITY Version 1.02”
- 2) 例えば，NEC ソフトウェア “ファイルコンバータ/8800” では，IBM 標準フォーマットのファイルと N88-DISK BASIC フォーマットのファイルとの相互変換を行う。
- 3) CP/M のファイルアロケーションマップの構造に関しては，例えば，村瀬康治著『応用 CP/M』アスキー出版局 (1982) p. 34 を見よ。
- 4) SHIMA, Masako, A. OKADA and F. WAKABAYASHI, unpublished data.
- 5) SHIMA, Masako, A. OKADA and F. WAKABAYASHI, unpublished data.

付 録

プログラムの使用法

I. “CONVRT. N88”

- ① “CONVRT. N88” をロードしランさせると，“Set PC-8801 DISKETTE on DRIVE 1, Set CP/M-86 DISKETTE on DRIVE 2, OK? (y/n)?” と表示される。
- ② プログラムがはいったディスクを抜き取り，別にデータファイル専用として PC-8801 (又は PC-9801) 用のディスクを用意して，ドライブ 1 にセットする。ドライブ 2 には，データのはいったマルチ 16 用のディスクをセットする。
- ③ ②が完了したら “y” を入力する。
- ④ 実行プログラムのメニューが表示されるので，その中から実行したいプログラムを選択し，その番号を入力する。

以下，各実行プログラムの操作法を説明する。

⑤ ディスクアロケーションマップ

データ読み取り，データ転送ではブロック番号でファイルを指定するので，各オリジナルディスクに対して，一度はこのプログラムを実行する必要がある。“Printer (p) or CRT (c)?” と聞いてくるので，プリンターに出力するか CRT に表示するかに応じて，それぞれ “p” 又は “c” を入力する。ディスクに記録されているファイルの名称と，そのファイルのディスク上での位置がブロック番号で出力される。

⑥ データ読み取り

マルチ16 ディスク内のデータを直接画面に表示する。表示させたいデータファイルの先頭ブロック番号とそのファイルのブロック数を入力する。1 ファイルが 2048 ch の場合，ブロック数は 5 である。

⑦ データ転送

- (1) プログラムメニューで “3” を入力すると “1. TN-1750 2. TN-7200?” と聞いてくるので，使用したマルチチャンネルアナライザーに応じて “1” 又は “2” を入力する。

- (2) 次に“データファイル Name?”と聞いてくる。CP/M-86 のデータファイル名(プライマリーネーム 8 文字まで、エクステンション 3 文字まで)では、PC 用(それぞれ、6 文字、3 文字)には使用できない場合があるので、新たに PC 用のファイル名を入力する。
- (3) 次に“BLOCK No.?”と聞いてくるので転送するデータファイルの先頭ブロック番号を入力する。PC-8801 では、約 1 分間で転送が終了する。
- ⑧ 1 つの実行プログラムが終了するとメニューに戻る。すべてを終了するときには“4”を入力する。

II. “RIDISPLAY”

- ① ドライブ 1 に“RIDISPLAY”と“READPRGM”の入ったディスクをセットする。ドライブ 2 には“CONVRT. N88”で作成した変換ディスク(ここではデータディスクと称す)をセットする。
- ② “RIDISPLAY”をロードし、ランさせる。この時、プリンターは電源を入れ、セレクト状態にしておく。

[データ読み込み]

- ③ 初期設定の後に“READPRGM”が自動的にロード・ランされ、“ヨミコミ 1. データブ 2. Key Board 3. Disk?”と聞いてくる。今の場合、ディスクから読み取るので“3”を入力する。
- ④ “Set DATA DISK on DRIVE 2 O.K.?(y/n)”と聞いてくるので、ドライブ 2 にデータディスクがセットされている時は“y”を入力する。
- ⑤ データディスクに記録されたファイル名を CRT 表示し、“データファイル Name?”と聞いてくるので目的のデータファイル名を入力する。
- ⑥ 次に使用したマルチチャンネルアナライザーを“1. TN-1750 2. TN-7200?”と聞いてくるので、“1”又は“2”を入力する。
- ⑦ 読み取りが開始され、CRT 上には読み取ったチャンネル番号が、TN-1750 では 8 ch ごと、TN-7200 では 14 ch ごとに表示される。
- ⑧ 読み取り終了後、“RIDISPLAY”を自動的にロード・ランし、スペクトル表示の準備をする。

[スペクトル表示]

- ⑨ 各ファンクションキーの機能は次のとおりである。
 - f1: XE……X軸の拡大・縮小率の変更
 - f2: YG……Y軸の拡大・縮小率の変更
 - f3: IPX……X軸の開始点の変更
 - f4: IPY……Y軸の開始点の変更
 - f5: SMT……3 点スムージングの実行
 - f6: セン……スペクトルをラインで表示
 - f7: テン……スペクトルを点で表示
 - f8: コピ……CRT 上のスペクトルをハードコピーする。
 - f9: PRT……数値データをプリンタ (p) 又は CRT (c) に出力する。
 - f10: END……終了
- ※ f1~f4 の設定変更を行ったあとは、f6 または f7 を実行する。
- ※ 新しいデータを扱う場合は一度終了させてから再びランさせる。

Program list 1: File converter "CONVRT. N88"

```

1000 'CONVRT.N88 ---- MULTI-16 diskette ヨミトリ・テンソク ----
1010 GOTO *MAIN
2000 *DATWRITE                               ' PC-8801 diskette ^ テンソク
2010 CLS
2020 INPUT " 1. TN-1750 OR 2. TN-7200 ";ITN
2030 INPUT "テータ ファイル Name "; DATNAME$
2040 INPUT " BLOCK No. "; BLOC
2050 OPEN DATNAME$ FOR OUTPUT AS #1
2060 PRINT #1 , "CP/M.86 Diskette -> PC-8801 Diskette " : CLOSE #1
      :PRINT " CP/M.86 Diskette -> PC-8801 Diskette テンソク チウ "
      :PRINT " シハラク オマチクダサイ ( About 1 minutes ) "
2070 OPEN DATNAME$ FOR APPEND AS #1
2080 FIELD #0, 64 AS A$(0),64 AS A$(1),64 AS A$(2),64 AS A$(3)
2090 FOR KK=1 TO 4
2100 GOSUB *DISALLOC : J=0
2120 FOR I=1 TO 16
2130 DUMMY$=DSKI$(2,SURF,CYLD,SECT)
2140 FOR K=0 TO 3:DAT$(J+K)=A$(K):NEXT
2150 SECT=SECT+1
2160 IF SECT=27 THEN SECT=1:SURF=SURF+1:IF SURF=2 THEN SURF=0:CYLD=CYLD+1
2170 J=J+4
2180 NEXT
2190 FOR JJ=0 TO 63:PRINT #1,DAT$(JJ);:NEXT
2200 BLOC=BLOC+1
2210 NEXT
2220 GOSUB *DISALLOC
2230 J=0
2240 IF ITN=2 THEN *TN7200
2250 FOR I=1 TO 2                               ' TN-1750
2260 DUMMY$=DSKI$(2,SURF,CYLD,SECT)
2270 FOR K=0 TO 3:DAT$(J+K)=A$(K):NEXT
2280 SECT=SECT+1
2290 IF SECT=27 THEN SECT=1:SURF=SURF+1:IF SURF=2 THEN SURF=0:CYLD=CLYD+1
2300 J=J+4
2310 NEXT
2320 FOR JJ=0 TO 4:PRINT #1,DAT$(JJ);:NEXT
2330 DATDUMMY$=LEFT$(DAT$(5),62)
2340 PRINT #1,DATDUMMY$;
2350 GOTO 2470
2360 *TN7200                               ' TN-7200
2370 FOR I=1 TO 7
2380 DUMMY$=DSKI$(2,SURF,CYLD,SECT)
2390 FOR K=0 TO 3:DAT$(J+K)=A$(K):NEXT
2400 SECT=SECT+1
2410 IF SECT=27 THEN SECT=1:SURF=SURF+1:IF SURF=2 THEN SURF=0:CYLD=CYLD+1
2420 J=J+4
2430 NEXT
2440 FOR JJ=0 TO 25:PRINT #1,DAT$(JJ);:NEXT
2450 DATDUMMY$=LEFT$(DAT$(26),48)
2460 PRINT #1,DATDUMMY$;
2470 CLOSE #1
2480 FOR J=1 TO 4:BEEP(1):FOR I=1 TO 250:NEXT:BEEP(0):FOR I=1 TO 200:NEXT:NEXT
2490 RETURN
3000 *DISALLOC                               ' フロック No. アンカン
3010 BLK=16*BLOC+53
3020 CYLD=INT(BLK/52)+1
3030 SECR=BLK-52*(CYLD-1)
3040 IF SECR<26 THEN SECT=SECR:SURF=0 ELSE SECT=SECR-26:SURF=1
3050 RETURN
4000 *DIR                                     ' ファイル コントロール フロック ヨミトリ
4010 INPUT "Printer (p) or CRT (c)";OT$
4020 IF OT$='p' OR OT$='P' THEN E$='LPT1:'
      ELSE IF OT$='c' OR OT$='C' THEN E$='SCRN:' ELSE BEEP:GOTO 4010
4030 OPEN E$ FOR OUTPUT AS #1

```

```

4040 FIELD #0,32 AS A$(0),32 AS A$(1),32 AS A$(2),32 AS A$(3),32 AS A$(4),
      32 AS A$(5),32 AS A$(6),32 AS A$(7)
4050 FOR SEC=1 TO 16
4060 DUMMY$=DSKI$(2,0,2,SEC)
4070 FOR I=0 TO 7
4080 IF LEFT$(A$(I),1)="▲" THEN 4160
4090 PRINT #1,MID$(A$(I),2,8);". ";MID$(A$(I),10,3);SPACE$(3);". ";
4100 FOR J=1 TO 16
4110 A$=MID$(A$(I),16+J,1)
4120 IF A$=CHR$(0) THEN 4140
4130 PRINT #1,USING "####";ASC(A$);
4140 NEXT
4150 PRINT #1,
4160 NEXT
4170 NEXT
4180 CLOSE #1
4190 RETURN
5000 *DATREAD
      CP/M.86 diskette テーマノヨミトリ
5010 CLS
5020 INPUT "BLOCK no. ";BLOC
5030 INPUT " How many blocks ";N
5040 INPUT "Printer (p) or CRT (c) ";OT$
5050 IF OT$="p" OR OT$="P" THEN E$="LPT1:"
      ELSE IF OT$="c" OR OT$="C" THEN E$="SCRN:" ELSE BEEP:GOTO 5040
5060 OPEN E$ FOR OUTPUT AS #1
5070 FOR KK=0 TO N-1
5080 GOSUB *DISALLOC
5090 FIELD #0,64 AS A$(0),64 AS A$(1),64 AS A$(2),64 AS A$(3)
5100 J=0
5110 FOR I=1 TO 16
5120 DUMMY$=DSKI$(2,SURF,CYLD,SECT)
5130 FOR K=0 TO 3:DAT$(J+K)=A$(K):NEXT
5140 SECT=SECT+1
5150 IF SECT=27 THEN SECT=1:SURF=SURF+1:IF SURF=2 THEN SURF=0:CYLD=CYLD+1
5160 FOR K=0 TO 3
5170 IF MID$(DAT$(J+K),2,1)>"▲" THEN PRINT #1,DAT$(J+K); ELSE 5230
5180 NEXT
5190 J=J+4
5200 NEXT
5210 BLOC=BLOC+1
5220 NEXT
5230 CLOSE #1
5240 RETURN
6000 *MAIN
      メインプログラム
6010 DEFINT A-Z
      インテライズ
6020 BLOC=0:BLK=0:CYLD=0:SECT=0:SECR=0:SURF=0:I=0:J=0:K=0:KK=0
      :A$=CHR$(0):OT$="c":E$="SCRN"
6030 DIM A$(7),DAT$(63)
6040 CONSOLE 0,25,0 : CLS
6050 PRINT "Set PC-8801 DISKETTE on DRIVE 1":
      PRINT "Set CP/M-86 DISKETTE on DRIVE 2"
      JOB Start
6060 INPUT " OK? (y/n) ";B$
6070 IF NOT(B$="y" OR B$="Y") THEN CLS:BEEP:GOTO 6050
6080 CLS
6090 PRINT "Functionヲ センタク シテクダサイ"
6100 INPUT " 1.ディスク アロケーション マップ 2.テーマノヨミトリ 3.テーマ テンソク 4.終了 ",K
6110 ON K GOSUB *DIR,*DATREAD,*DATWRITE,*EN
6120 CLS : GOTO 6090
7000 *EN CONSOLE 0,25,1 : CLS
      シュリヨク プログラム
7010 END

```

Program list 2: Spectrum display program "RIDISPLAY"

```

10000 ˆ RIDISPLAY
10010  GOTO *MAIN
11000  *SMOOT      : CLS 2:C=0                ˆ 3 points スム-シ`ク`
11010  FOR I=1 TO N-2:C1=C(I):C(I)=(C+C(I)+C(I+1))/3:C=C1:NEXT
11020  GOSUB *SEN
11030  RETURN
11100  *SEN        : CLS 2                    ˆ レソ`ク` セン
11110  POINT (PX-1,-IPY+1)
11120  IF EP<EPX THEN EPP=EP ELSE EPP=EPX
11130  IF XE<2 THEN FOR I=IPX TO EPP:LINE -(I,-C(I)*FK):NEXT
        ELSE FOR I=IPX TO EPP:LINE (I,IPY)-(I,-C(I)*FK):NEXT
11140  RETURN
11200  *TEN        : CLS 2                    ˆ ホ`イント` フ`ロ`ット
11210  IF EP<EPX THEN EPP=EP ELSE EPP=EPX
11220  FOR I=IPX TO EPP:PSET (I,-C(I)*FK):NEXT
11230  RETURN
11300  *XE         ˆ X `シ`ク` カクタ`イ
11310  LOCATE 14,24:PRINT SPACE$(8);:LOCATE 10,24:COLOR 6:INPUT "XE =";XE
11320  COLOR 0 : GOSUB *KOMENT
11330  RETURN
11400  *YG         ˆ Y `シ`ク` カクタ`イ
11410  LOCATE 26,24:PRINT SPACE$(9);:LOCATE 22,24:COLOR 6:INPUT "YG =";YG
11420  COLOR 0 : GOSUB *KOMENT
11430  RETURN
11500  *IPX        ˆ X `シ`ク` イニシ`アル` ホ`イント
11510  LOCATE 7,23:PRINT SPACE$(4):LOCATE 2,23:COLOR 6:INPUT "IPX=";IPX
11520  COLOR 0:LOCATE 0,23:PRINT SPACE$(10) : GOSUB *KOMENT
11530  RETURN
11600  *IPY        ˆ Y `シ`ク` イニシ`アル` ホ`イント
11610  LOCATE 3,22:PRINT SPACE$(6):LOCATE 2,22:COLOR 6:INPUT "IPY=";IPY
11620  COLOR 0:LOCATE 0,22:PRINT SPACE$(10) : GOSUB *KOMENT
11630  RETURN
12000  *KOMENT     : CLS                    ˆ コ`メント
12010  IF XE=0 THEN XE=1
12020  IF YG=0 THEN YG=1
12030  EPX=IPX+500/XE:EPY=IPY+350*KF/YG : Z1=(EPX-IPX)/5:Z2=(EPY-IPY)/3.5
12040  LOCATE 10,24:COLOR 4:PRINT "XE =";XE;:COLOR 0:PRINT SPC(21-POS(0));
12050  LOCATE 22,24:COLOR 4:PRINT "YG =";YG;:COLOR 0:PRINT SPC(37-POS(0));
12060  LOCATE 7,23:PRINT USING "####" ;IPX;
12070  LOCATE 20,23:PRINT USING "####" ;IPX+ Z1;
12080  LOCATE 33,23:PRINT USING "####" ;IPX+2*Z1;
12090  LOCATE 45,23:PRINT USING "####" ;IPX+3*Z1;
12100  LOCATE 58,23:PRINT USING "####" ;IPX+4*Z1;
12110  LOCATE 70,23:PRINT USING "####" ;EPX
12120  LOCATE 3, 3:PRINT USING "#####";IPY+3*Z2;
12130  LOCATE 3,10:PRINT USING "#####";IPY+2*Z2;
12140  LOCATE 3,16:PRINT USING "#####";IPY+ Z2;
12150  LOCATE 3,22:PRINT USING "#####";IPY;
12160  WINDOW (IPX,-EPY)-(EPX,-IPY)
12170  LOCATE 74, 1:PRINT "f1:XE" : LOCATE 74, 2:PRINT "f2:YG"
12180  LOCATE 74, 3:PRINT "f3:IPX" : LOCATE 74, 4:PRINT "f4:IPY"
12190  LOCATE 74, 5:PRINT "f5:SMT" : LOCATE 74, 7:PRINT "f6:セン"
12200  LOCATE 74, 8:PRINT "f7:テン" : LOCATE 74, 9:PRINT "f8:コ`ピ`"
12210  LOCATE 74,10:PRINT "f9:PRT" : LOCATE 74,11:PRINT "f10:EN"
12220  LOCATE 0, 0:PRINT DATNAME$
12230  LOCATE 38,24:PRINT "Live Time =";:PRINT USING "#####";LT;:PRINT "sec";
12240  LOCATE 60,24:PRINT "Dead Time =";:PRINT USING "###.##";DED;:PRINT "%";
12250  RETURN
12500  *COORD      : CLS                    ˆ サ`ヒョウ`シ`ク` カキ`コ`ミ
12510  ON KEY GOSUB *XE,*YG,*IPX,*IPY,*SMOOT,*SEN,*TEN,*REG,*DATTYPE,*EN:KEY ON
12520  FOR I= 80 TO 580 STEP 10 : LINE (I,359)-(I,363) : NEXT
12530  FOR I= 80 TO 580 STEP 50 : LINE (I,363)-(I,367) : NEXT
12540  FOR I= 9 TO 359 STEP 10 : LINE ( 75,I)-( 79,I) : NEXT
12550  FOR I= 59 TO 359 STEP 50 : LINE ( 71,I)-( 75,I) : NEXT

```

```

12560 VIEW (80, 9)-(580,359),,4
12570 RETURN
13000 *REG
13010 FOR I=1 TO 11:LOCATE 74,I:PRINT "          ' CRT ハート コピ-
          *:NEXT I
13020 COPY 3
13030 GOSUB *KOMENT
13040 RETURN
14000 *DATTYPE
          ' テータ フォント or CRT シュリョク
14010 VIEW (0,0)-(639,399) : WINDOW(0,0)-(639,399) : CLS 3
14020 INPUT "Printer (p) or CRT (c)";OT$
14030 IF OT$="p" OR OT$="P" THEN E$="LPT1:";LPRINT CHR$(27);"N";CHR$(27);"J";
          ELSE IF OT$="c" OR OT$="C" THEN E$="SCRN:" ELSE BEEP:GOTO 14020
14040 OPEN E$ FOR OUTPUT AS #1
14050 PRINT #1, DATNAME$ : PRINT #1,
14060 PRINT #1, "LT=" ; : PRINT #1,USING "#####";LT;
14070 PRINT #1, " RT=" ; : PRINT #1,USING "#####";RT;
14080 PRINT #1, " Dead Time=" ; : PRINT #1,USING "###.###" ;DED;
14090 PRINT #1, "%" : PRINT #1,
14100 FOR I=0 TO 255 : IJ=8*I
14110 PRINT #1,USING "#####";IJ;
14120 FOR J=0 TO 7:PRINT #1,USING "#####";C(IJ+J);:NEXT
14130 PRINT #1,
14140 NEXT
14150 IF OT$="p" OR OT$="P" THEN LPRINT CHR$(27);">";
14160 CLOSE #1
14170 GOSUB *COORD : GOSUB *KOMENT
14180 RETURN
15000 *MAIN
          ' メイン フォグラム
15010 WIDTH 80,25 : CONSOLE 0,25,0,0 : SCREEN 2 : CLS : LPRINT CHR$(27);">";
15020 DEFINT E,H-N: DEFSNG A-D,F-G,0-Z
15030 A=0:I=0:J=0:K=0:N=0:IJ=0:IP=0:EP=0:CMAx=0:XE=1:YG=1:IPX=0:IPY=0:EPX=0
          :EPY=0:EPP=0:LT=0:RT=0:DED=0:FK=1:KF=1:X=0:Z1=0:Z2=0:C=0:C1=0:ITN=1
15040 A$="":EL$="":DT$="":TM$="":DATNAME$="":OT$="C":E$="SCRN:"
15050 CHAIN "READPRGM",20000,ALL
          ' テータ ヨミコミ
          ' サブプログラム
          ' Display
15060 GOSUB *COORD
15070 GOSUB *KOMENT
15080 ON ERROR GOTO *FUKKI
15090 *LOOP : GOTO *LOOP
16000 *FUKKI
          ' エラー ショリ
16010 BEEP:VIEW (0,0)-(639,399):WINDOW (0,0)-(639,399):CLS 3:BEEP
          :IF ERR=6 THEN LOCATE 20, 5 : PRINT "オハ-70- テース。セツタイ ヲオシテ クタサイ。"
          ELSE LOCATE 20, 5 : PRINT "エラー カン ショクシマシタ。モウ イチ!"
16020 FOR I=0 TO 2500 :NEXT : RESUME 15060
17000 *EN
          ' シュリョク フォグラム
17010 KEY OFF : ON ERROR GOTO 0
17020 VIEW (0,0)-(639,399) : WINDOW (0,0)-(639,399) : CONSOLE 0,25,1,0 : CLS 3
          : LPRINT CHR$(27);"H" ; CHR$(27) ; "J" ;
17030 END

```

Program list 3: Data input program "READPRGM" for "RIDISPLAY"

```

20000 'READPRGM
20010 GOTO *MAIN
21000 *READDISK ' diskette テーマ ヨミトリ
21010 PRINT 'Set DATA DISK on DRIVE 2'
21020 INPUT ' O,K.? (y/n) ', B$
21030 IF NOT(B$="y" OR B$="Y") THEN CLS:BEEP:GOTO 21010
21040 CLS : IP=0:EP=2047:N=2048
21050 FILES 2
21060 INPUT ' テーマ ファイル NAME '; DATNAME$
21070 DATNAME1$="2:"+DATNAME$
21080 INPUT ' 1. TN-1750 or 2. TN-7200 '; ITN
21090 OPEN DATNAME1$ FOR INPUT AS #1
21100 IF ITN=2 THEN *TN7200 ELSE IF NOT(ITN=1) THEN BEEP:GOTO 21080
21110 FOR J=1 TO 14:INPUT #1,A:NEXT ' TN-1750
21120 INPUT #1,LT,RT:IF LT=0 THEN DAD=0 ELSE DED=(RT-LT)/LT*100
21130 FOR J=17 TO 23:INPUT #1,A:NEXT
21140 FOR I=0 TO 255:I=8*I:INPUT #1,A,C(IJ),C(IJ+1),C(IJ+2),C(IJ+3),
C(IJ+4),C(IJ+5),C(IJ+6),C(IJ+7):PRINT USING "#####";A;:NEXT
21150 GOTO 21220
21160 *TN7200 ' TN-7200
21170 FOR J=1 TO 10:INPUT #1, A:NEXT
21180 INPUT #1,LT,A,RT:IF LT=0 THEN DED=0 ELSE DED=(RT-LT)/LT*100
21190 FOR J=14 TO 23:INPUT #1,A:NEXT
21200 FOR I=0 TO 145:IJ=14*I:INPUT #1,A,C(IJ),C(IJ+1),C(IJ+2),C(IJ+3),
C(IJ+4),C(IJ+5),C(IJ+6),C(IJ+7),C(IJ+8),C(IJ+9),C(IJ+10),C(IJ+11),
C(IJ+12),C(IJ+13),B:PRINT USING "#####";A;:NEXT
21210 INPUT #1,A,C(2044),C(2045),C(2046),C(2047):PRINT USING "#####";A
21220 CLOSE #1
21230 RETURN
22000 *READKEYB ' キーボード から ヨミトリ
22010 ON HELP GOSUB *TEISEI : HELP ON
22020 PRINT ' テイセイ スル トキ ハ HELP KEY ラ オシテ クタサイ '
22030 INPUT ' Initial Point '; IP
22040 INPUT ' End Point '; EP
22050 N=EP-IP+1
22060 PRINT ' テーマ ラ イレテ クタサイ !! '
22070 FOR I=IP TO EP
22080 PRINT ' Ch. No. ',I ; : INPUT ' Counts= ', C(I)
22090 NEXT
22100 HELP OFF
22110 INPUT ' テイセイ ヲマスカ (y/n) '; A$
22120 IF A$="y" OR A$="Y" THEN GOSUB *TEISEI
22130 IF NOT (A$="n" OR A$="N") THEN BEEP : GOTO 22110
22140 RETURN
23000 *READDATA ' テーマ フォン から ヨミトリ
23010 RESTORE
23020 READ EL$,DT$,TM$,MT,DED,IP,EP
23030 N=EP-IP+1
23040 FOR I=IP TO EP:READ C(I):NEXT
23050 RETURN
24000 *MAX ' C(I) ノ サイダイチ
24010 FOR I=IP TO EP:IF CMAX<C(I) THEN CMAX=C(I):NEXT
24020 KF=INT(CMAX/32768!)+1 : FK=1/KF
24030 RETURN
24200 *TEISEI ' テイセイ プログラム
24210 PRINT : INPUT ' テイセイ スル Ch. No. '; I
24220 INPUT ' Counts= ', C(I)
24230 RETURN
25000 *MAIN ' メイン プログラム
25010 DIM C(2047)
25020 ON ERROR GOTO *FUKKI
25030 PRINT ' ----- ニュウリョク ----- '
25040 INPUT ' ヨミヨミ 1. テーマ フォン 2. Key Board 3. Disk '; K
25050 IF K<1 OR K>3 THEN BEEP : CLS : PRINT ' トウ イチト!!!! ' : GOTO 25020

```

```
25060 ON K GOSUB *READDATA,*READKEYB,*READDISK
25070 GOSUB *MAX
25080 CLS
25090 CHAIN "RIDISPLAY",15060,ALL
25100 END
26000 *FUKKI
26010 CLS : BEEP
26020 IF ERR=53 THEN PRINT "リテイサレタ ファイル カ" アリマセン。";
      ELSE PRINT "エラー カ" オキマシタ。";
26030 PRINT "      モウ イチト!" : PRINT : PRINT
26040 RESUME 25030
```